



INTERNAL DOCUMENT No. 299

**Digiquartz pressure transmitter software used
in pressure calibration at IOSDL**

S Keene

1991

**INSTITUTE OF OCEANOGRAPHIC SCIENCES
DEACON LABORATORY**

INTERNAL DOCUMENT No. 299

**Digiquartz pressure transmitter software used
in pressure calibration at IOSDL**

S Keene

1991

Wormley
Godalming
Surrey GU8 5UB
Tel 0428 684141
Telex 858833 OCEANS G
Telefax 0428 683066

DOCUMENT DATA SHEET

AUTHOR KEENE, S	PUBLICATION DATE 1991
TITLE Digiquartz pressure transmitter software used in pressure calibration at IOSDL.	
REFERENCE Institute of Oceanographic Sciences Deacon Laboratory, Internal Document, No. 299, 23pp. (Unpublished manuscript)	
ABSTRACT Software has been developed at IOSDL to support the Paroscientific Pressure Transmitter Model 740 when coupled with the Budenberg Dead Weight Tester. This software is now incorporated into the pressure calibration facility at IOSDL to work on IBM and BBC computers.	
KEYWORDS BUDENBERG DEAD WEIGHT TESTER DIGIQUARTZ PRESSURE	
ISSUING ORGANISATION Institute of Oceanographic Sciences Deacon Laboratory Wormley, Godalming Surrey GU8 5UB. UK. Director: Colin Summerhayes DSc Telephone Wormley (0428) 684141 Telex 858833 OCEANS G. Facsimile (0428) 683066	
Copies of this report are available from: The Library, PRICE £00.00	

Contents	Page
Section 1 Introduction	7
Section 2 Aims	7
Section 3 Software Programs	8
Section 4 Data Configuration	9
Section 5 Connections	10
Section 6 Conclusion	10
Section 7 Further information	10
Section 8 Acknowledgements	10
Section 9 References	10

Diagrams and Program Listings

Fig. 1 Software flowchart	11
Fig. 2 BBC cable connections	12
Fig. 3 IBM cable connections	13
Listing 1 IBM program listing	14
Listing 2 BBC program listing	18

Digiquartz Pressure Transmitter Software used in Pressure Calibrations at IOSDL.

1. Introduction

The Paroscientific Series 1000 Digiquartz Intelligent Transmitter Model 740-10K provides temperature corrected pressure outputs, via an RS232 link, to a computer or a stand-alone readout display. The Transmitter has a range of 0-6000 psi, can sample in steps of approximately 3ms to 30s and is accurate to 0.01%.

During CTD calibrations, when connected to the Budenberg Pressure Standard, the Transmitter enables accurate pressure readings and a means of double checking the Budenberg to be obtained.

The manual supplied with the Digiquartz 740 (ref.1) includes all the operating commands to control and calibrate the Transmitter. The software supplied with the Digiquartz 740 was compatible with IBM PC's only and was unsuitable for calibration purposes. It therefore was necessary to write a complete set of software to include both the IBM PC and the BBC Microcomputer. This report provides an explanation of that software.

2. Aims

2.1 CTD calibrations require pressure to be exerted on the CTD pressure cell, this is done by the Budenberg Dead Weight Tester which forces fluid against the pressure cell. The Transmitter is situated inbetween the Budenberg and the CTD cell and displays the pressures.

The CTD transmits its pressure readings to a BBC for display; thus the first aim is to transmit pressure readings from the Digiquartz 740 to a computer for display, and also to average these readings every twelve or so transmissions to aid in legibility.

2.2 The Digiquartz 740 Transmitter has a facility enabling the user to choose either, one of the preprogrammed pressure engineering units or to define a unit. Because the unit used for calibrations, dbars, was not one of the preprogrammed units it was necessary to install it as the user defined unit. The next aim is to write a program to make the changing of pressure units more accessible for the user.

2.3 The Transmitter has an internal temperature sensor which is used for the thermal compensation of pressure; the third aim is to write some software to display the temperature the Transmitter is reading.

2.4 The last aim is to write a 'menu' type program to combine all the programs together therefore achieving a neat self-contained package.

3. Software Programs

The software comprises of two discs of programs, one for the IBM PC and one for the BBC Microcomputer. The IBM software is written in IBM Basic 3.10 and the BBC software is written in BBC Basic. Both discs contain the same filenames and perform the same tasks. Detailed listings of the programs are found at the end of this report.

Programs:

both discs)

3.1 PRES - This program commands the Digiquartz 740 to continuously transmit pressure readings to the computer. These readings are displayed on the screen and averaged every twelve transmissions (thirteen on the IBM); the averaged reading and the previously averaged reading are also displayed. This program satisfies aim 1.

3.2 PRESREM - This program is the same as PRES but has detailed REM lines to describe the workings of the program.

3.3 UNIT - This displays on the screen the nine different pressure units the Digiquartz 740 can display and the unit the Transmitter is currently displaying. The user can choose the pressure unit required, this chosen unit will be permanently written in the Transmitter's eeprom until another unit is chosen. To keep eeprom wear to a minimum this facility is best used only when necessary. This program satisfies aim 2.

3.4 UNITREM - As UNIT but with added REM lines in order to explain how the program works.

3.5 TEMP - This program commands the Digiquartz 740 to transmit temperature commands, these temperatures readings are displayed on the screen in degrees centigrade. Whilst the Transmitter is sending temperature readings it displays "-----" to show that the display is not being updated. This program satisfies aim 3.

3.6 **TEMPREM** - This is the same as **TEMP** but with added **REM** statement to aid in the legibility of how the program operates.

3.7 **MENU** - This program lets the user choose between the following programs: **PRES**, **UNIT**, **TEMP** or to 'exit' the software. **MENU** on the BBC is automatically loaded by inserting the disc and pressing **SHIFT-BREAK**; on the IBM, **MENU** must be loaded and run from Basic.

The ----**REM** programs should not be used as working programs as they are slower than the other programs, they are only included as a guide through the programs commands.

4. Data Configuration

Baud rate : 1200 (user definable from 150-19200).

Data framing: 8 data bits, 1 stop bit, no parity.
Other user selectable choices:
7 data bits, 1 stop bit, odd parity.
7 data bits, 1 stop bit, even parity.

Handshaking : Not required.

Input level : Any RS232 compatible level.

Output level: +/- 5 vdc nominal.

Pressure User selectable:
Units: psi, mbar, bar, Kpa, Mpa,
inches of Hg, torr, meters of water,
or user definable (dbar).

5. Connections

Connections are:

The Digiquartz 740 uses a 9-way D-type port.

The IBM PC uses the 25-way D-type serial communications port.

The BBC Microcomputer uses the RS 423 serial port.

DESCRIPTION	DIGIQUARTZ	IBM	BBC	
	Pin no.	Pin no.	Pin no.	
Chassis ground.	1	1	-	
Data to comp.	2	3	4	(On the BBC
Data from comp.	3	2	3	pins 2 and 5
Signal ground	5	7	1	are linked)

See figures 2 and 3 for further details.

6. Conclusion

Because the sampling time of the Digiquartz Transmitter for calibration purposes can be set at a fairly slow rate, (0.8 samples/second was chosen for good accuracy), it was possible to write the software in Basic rather than a faster, more hostile language. However, if in the future faster sampling speeds are required, (up to 120 samples/second are possible), programs in a quicker language will have to be written.

The software appears to be running as hoped and all previously stated aims have been met, but at this present moment in time no calibrations have been taken using it.

7. Further Information

The manual supplied with the Digiquartz 740-10K Transmitter (ref. 1) includes all the operating commands to control and calibrate the instrument.

8. Acknowledgements

Thanks are due to Pat Gwilliam for the helpful suggestions.

9. References

Ref. 1 - 1000 series Digiquartz Intelligent Transmitter Programming and Operation Manual.
Copyright Paroscientific Incorporated. Paroscientific document no. 8107-001.





Fig. 1
SOFTWARE FLOWCHART

DESCRIPTION	DIGIQUARTZ	BBC
	Pin no.	Pin no.
Chassis ground	1	-
Data to comp.	2	4
Data from comp.	3	3
Signal ground	5	1

(On the RS423 pins 2 and 5 are linked).




As viewed from the rear.

FIG. 2	DIGIQUARTZ 9-WAY D-TYPE PORT TO BBC RS423 PORT.	DRAWN	S.KEENE
		DATE	31/1/91

INSTITUTE OF OCEANOGRAPHIC SCIENCES DEACON LABORATORY.

DESCRIPTION	DIGIQUARTZ	IBM PC
	Pin no.	Pin no.
Chassis ground	1	1
Data to comp.	2	3
Data from comp.	3	2
Signal ground	5	7



As viewed from the rear.

FIG. 3	DIGIQUARTZ 9-WAY D-TYPE PORT TO IBM 25-WAY COMMUNICATIONS ADAPTOR.	DRAWN	S. KEENE
		DATE	31/1/91

INSTITUTE OF OCEANOGRAPHIC SCIENCES DEACON LABORATORY.

```
10 REM -----MENU-----
20 REM -----17-1-91-S.KEENE
30 REM Menu program for the digiquartz
40 REM standard in IBM basic 3.10
50 REM*
60 REM*
70 SCREEN 0,1:CLS:COLOR 3,0
80 LOCATE 4,30:PRINT"MENU":COLOR 4,0
90 LOCATE 8,20:PRINT"1.....UNITS CHANGING PROGRAM"
100 LOCATE 11,20:PRINT"2.....PRESSURE AVERAGING PROGRAM"
110 LOCATE 14,20:PRINT"3.....TEMPERATURE SAMPLING PROGRAM"
120 LOCATE 17,20:PRINT"4.....EXIT"
130 Q$=INKEY$:IF Q$<>""THEN GOTO 130
140 Q$=INKEY$:IF Q$=""THEN GOTO 140
150 COLOR 3,0:LOCATE 20,30:PRINT"WAIT":COLOR 4,0
160 IF Q$="1"THEN CHAIN"UNIT"
170 IF Q$="2" THEN CHAIN"PRES"
180 IF Q$="3"THEN CHAIN"TEMP"
190 IF Q$="4" THEN CLOSE#1:CLS:COLOR 4,0:END
200 GOTO 70
```

```
10 REM          ----UNITREM----
20 REM          17/1/91-S.KEENE
30 REM A program to change the pressure
40 REM units of the Digiquartz standard
50 REM via the communication adaptor#1.
60 REM   Written in IBM basic 3.10.
70 REM Use UNIT as the working program.
80 REM*
90 REM*
100 REM Clear the communication adaptor
110 OPEN"COM1:1200,N,8,1,CS,DS,LF"AS#1:PRINT#1,"*0100VR":CLOSE#1
120 REM Tell Digiquartz to transmit current pressure unit
130 OPEN"COM1:1200,N,8,1,CS,DS,LF"AS#1:PRINT#1,"*0100UN"
140 INPUT#1,G$:CLOSE#1
150 CLS:LOCATE 2,24:PRINT"THE STANDARD IS DISPLAYING"
160 LOCATE 2,52:COLOR 3,0
170 IF G$="*0001UN=1"THEN PRINT"psi"
180 IF G$="*0001UN=2"THEN PRINT"mbar or hPa"
190 IF G$="*0001UN=3"THEN PRINT"bar"
200 IF G$="*0001UN=4"THEN PRINT"kPa"
210 IF G$="*0001UN=5"THEN PRINT"MPa"
220 IF G$="*0001UN=6"THEN PRINT"in Hg"
230 IF G$="*0001UN=7"THEN PRINT"mm Hg or torr"
240 IF G$="*0001UN=8"THEN PRINT"m H"CHR$(253)"O"
250 IF G$="*0001UN=0"THEN PRINT"dbar"
260 COLOR 4,0:LOCATE 4,19:PRINT"ENTER THE NUMBER OF THE UNIT REQUIRED"
270 LOCATE 6,4:PRINT"1.....psi":LOCATE 7,4:PRINT"2.....mbar or hPa"
280 LOCATE 8,4:PRINT"3.....bar":LOCATE 9,4:PRINT"4.....kPa"
290 LOCATE 10,4:PRINT"5.....MPa":LOCATE 11,4:PRINT"6.....in Hg"
300 LOCATE 12,4:PRINT"7.....mm Hg or torr"
310 LOCATE 13,4:PRINT"8.....m H"CHR$(253)"O"
320 LOCATE 14,4:PRINT"9.....dbar"
330 LOCATE 15,4:PRINT"0.....to exit to menu"
340 LOCATE 10,40
350 REM Wait for user input
360 K$=INKEY$:IF K$<>""THEN GOTO 360
370 K$=INKEY$:IF K$=""THEN GOTO 370
380 LOCATE 20,30:COLOR 3,0:PRINT"WAIT":COLOR 4,0
390 REM Open communications adaptor to change pressure units
400 OPEN"COM1:1200,N,8,1,CS,DS,LF"AS#1
410 IF MID$(K$,1,1)="1"THEN PRINT#1,"*0100EW*0100UN=1"
420 IF MID$(K$,1,1)="2"THEN PRINT#1,"*0100EW*0100UN=2"
430 IF MID$(K$,1,1)="3"THEN PRINT#1,"*0100EW*0100UN=3"
440 IF MID$(K$,1,1)="4"THEN PRINT#1,"*0100EW*0100UN=4"
450 IF MID$(K$,1,1)="5"THEN PRINT#1,"*0100EW*0100UN=5"
460 IF MID$(K$,1,1)="6"THEN PRINT#1,"*0100EW*0100UN=6"
470 IF MID$(K$,1,1)="7"THEN PRINT#1,"*0100EW*0100UN=7"
480 IF MID$(K$,1,1)="8"THEN PRINT#1,"*0100EW*0100UN=8"
490 IF MID$(K$,1,1)="9"THEN PRINT#1,"*0100EW*0100UN=0"
500 IF MID$(K$,1,1)="9"THEN PRINT#1,"*0100EW*0100UF=.6894757"
510 IF MID$(K$,1,1)="0"THEN CLOSE#1:CLEAR:CHAIN"MENU"
520 REM Close communications adaptor#1
530 CLOSE#1
540 GOTO 130
```

```
10 REM          ----PRESREM----
20 REM          17/1/91-S.KEENE
30 REM A program to average the pressure
40 REM transmitted from the DigiQuartz
50 REM standard via communication port#1
60 REM          using IBM basic 3.10.
70 REM Use PRES as the working program.
80 REM*
90 REM*
100 DIM V$(12):F=0:E=0:S=0
110 SCREEN 0,1:CLS
120 REM Clear communication adaptor#1.
130 OPEN"COM1:1200,N,8,1,CS,DS,LF"AS#1:PRINT#1,"*0100VR":CLOSE#1
140 REM Open communication adaptor#1.
150 OPEN"COM1:1200,N,8,1,CS,DS,LF"AS#1
160 REM Tell DigiQuartz standard to transmit current pressure units.
170 PRINT#1,"*0100UN":INPUT#1,W$
180 COLOR 4,0:LOCATE 3,25:PRINT"PRESSURE IN":LOCATE 3,38:COLOR 3,0
190 IF W$="*0001UN=1"THEN PRINT "psi"
200 IF W$="*0001UN=2"THEN PRINT "mbar or hPa"
210 IF W$="*0001UN=3"THEN PRINT "bar"
220 IF W$="*0001UN=4"THEN PRINT "kPa"
230 IF W$="*0001UN=5"THEN PRINT "MPa"
240 IF W$="*0001UN=6"THEN PRINT "in Hg"
250 IF W$="*0001UN=7"THEN PRINT "mm Hg or torr"
260 IF W$="*0001UN=8"THEN PRINT "m H"CHR$(253)"O"
270 IF W$="*0001UN=0"THEN PRINT "dbar"
280 REM Clear communication adaptor#1
290 PRINT#1,"*0100VR":CLOSE#1
300 COLOR 4,0:LOCATE 7,5:PRINT"PRESSURE"
310 LOCATE 7,23:PRINT"LAST AVERAGE:"
320 LOCATE 7,48:PRINT"DIFFERENCE IN AVERAGES"
330 LOCATE 9,23:PRINT"THIS AVERAGE:"
340 LOCATE 18,20:PRINT"PRESS ANY KEY TO RETURN TO MENU":COLOR 3,0
350 REM Tell DigiQuartz standard to continuously transmit pressure.
360 OPEN"COM1:1200,N,8,1,CS,DS,LF"AS#1:PRINT#1,"*0100P4"
370 INPUT#1,V$(A):V$(A)=MID$(V$(A),6):IF A=12 THEN GOTO 440
380 IF INKEY$<>""THEN GOTO 400
390 LOCATE 9,6:PRINT"          ":LOCATE 9,6:PRINT V$(A):A=A+1:GOTO 370
400 LOCATE 20,30:COLOR 3,0:PRINT"WAIT"
410 REM Close communications adaptor and return to the menu.
420 CLOSE#1:COLOR 4,0:CHAIN"MENU"
430 REM 440-480 averaging the pressures.
440 FOR A=0 TO 12
450 D$=LEFT$(V$(A),6):N=VAL(D$):S=S+N:NEXT:C$=STR$(S/13)
460 D=VAL(C$):LOCATE 7,37:PRINT"          ":LOCATE 7,37:PRINT E
470 F=D-E:E=D:LOCATE 9,37:PRINT"          ":LOCATE 9,37:PRINT D
480 LOCATE 9,55:PRINT"          ":LOCATE 9,55:PRINT F
490 REM Clear the communication adaptor#1.
500 PRINT#1,"*0100VR":CLOSE#1:S=0:A=0:GOTO 360
```

```
10 REM      ----TEMPREM---
20 REM      17/1/91-S.KEENE
30 REM A program to sample the temperature
40 REM with the Digiquartz, using the
50 REM communications adaptor#1; written
60 REM      in IBM basic.
70 REM Use TEMP as the working program.
80 REM*
90 REM*
100 ON ERROR GOTO 250
110 CLS:LOCATE 8,25:COLOR 4,0:PRINT"TEMPERATURE IN DEGREES C."
120 REM Clear the communication adaptor.
130 OPEN"COM1:1200,N,8,1,CS,DS,LF"AS#1:PRINT#1,"*O100VR":CLOSE#1
140 REM Transmit the temperature command.
150 OPEN"COM1:1200,N,8,1,CS,DS,LF"AS#1:PRINT#1,"*O100Q4"
160 INPUT#1,T$:T$=MID$(T$,6)
170 LOCATE 13,35:COLOR 3,0:PRINT T$
180 COLOR 4,0:LOCATE 18,23:PRINT"PRESS ANY KEY TO RETURN TO MENU"
190 IF INKEY$<>""THEN GOTO 220
200 GOTO 160
210 REM Clear the communication adaptor and return to the menu.
220 LOCATE 20,36:COLOR 3,0:PRINT"WAIT":PRINT#1,"*O100VR"
230 CLOSE#1:CLEAR:CHAIN"MENU"
240 REM Clear the communication adaptor.
250 PRINT#1,"*O100VR":CLOSE#1
```

```
> LIST
 10 REM      ----MENU-----
 20 REM      17/1/91-S.KEENE
 30 REM      Menu program for
 40 REM      the Digiquartz standard
 50 REM      in BBC basic.
 60 REM.
 70 REM.
 80 MODE7
 90 VDU 23,1,0;0;0;0;0;
100 CLS
110 PRINT TAB(13,2);CHR$(132);CHR$(141);"MENU"
120 PRINT TAB(13,3);CHR$(132);CHR$(141);"MENU"
130 PRINT TAB(0,7);CHR$(129);"1.....UNITS CHANGING PROGRAM"
140 PRINT TAB(0,10);CHR$(129);"2.....PRESSURE AVERAGING PROGRAM"
150 PRINT TAB(0,13);CHR$(129);"3.....TEMPERATURE SAMPLING PROGRAM"
160 PRINT TAB(0,16);CHR$(129);"4.....EXIT"
170 A$=GET$
180 PRINT TAB(13,20);CHR$(130);"WAIT"
190 IF A$="1"THEN CHAIN"UNIT"
200 IF A$="2"THEN CHAIN"PRES"
210 IF A$="3"THEN CHAIN"TEMP"
220 IF A$="4"THEN CLS:END
230 GOTO 100
>VDU3
```

> LIST

```
10 REM      ---UNITREM---
20 REM      17/1/91-S.KEENE
30 REM      Changing the pressure units
40 REM      of the Digiquartz standard
50 REM      via the RS423 port.
60 REM      Using BBC basic.
70 REM      Use "UNIT" as the
80 REM      working program
90 REM.
100 REM.
110 REM 1200 Baud receive
120 *FX7,4
130 REM 1200 Baud transmit
140 *FX8,4
150 REM Flush all buffers
160 *FX15,0
170 REM Enable RS423 as output device
180 *FX3,7
190 REM Clear Digiquartz command
200 PRINT"*0100VR"
210 REM Flush all buffers
220 *FX15,0
230 REM Enable RS423 as input device
240 *FX2,1
250 REM Printer sink to buffer input
260 *FX5,0
270 REM Show pressure units command
280 PRINT"*0100UN"
290 REM Printer as output buffer
300 *FX3,2
310 INPUT A$
320 REM Enable screen as output
330 *FX3,4
340 REM Enable keyboard as input
350 *FX2,2
360 MODE7
370 CLS
380 PRINT TAB(2,5);CHR$(129);"THE STANDARD IS DISPLAYING"
390 IF A$="*0001UN=1"THEN PRINT TAB(29,5);CHR$(134);"psi"
400 IF A$="*0001UN=2"THEN PRINT TAB(29,5);CHR$(134);"mbar"
410 IF A$="*0001UN=3"THEN PRINT TAB(29,5);CHR$(134);"bar"
420 IF A$="*0001UN=4"THEN PRINT TAB(29,5);CHR$(134);"kPa"
430 IF A$="*0001UN=5"THEN PRINT TAB(29,5);CHR$(134);"MPa"
440 IF A$="*0001UN=6"THEN PRINT TAB(29,5);CHR$(134);"in Hg"
450 IF A$="*0001UN=7"THEN PRINT TAB(29,5);CHR$(134);"mm Hg"
460 IF A$="*0001UN=8"THEN PRINT TAB(29,5);CHR$(134);"m H2O"
470 IF A$="*0001UN=0"THEN PRINT TAB(29,5);CHR$(134);"dbar"
480 PRINT TAB(0,8);CHR$(129);"ENTER THE NUMBER OF THE UNITS REQUIRED"
490 PRINT TAB(0,10);CHR$(129);"1...psi"
500 PRINT TAB(0,11);CHR$(129);"2...mbar or hPa"
510 PRINT TAB(0,12);CHR$(129);"3...bar"
520 PRINT TAB(0,13);CHR$(129);"4...kPa"
530 PRINT TAB(0,14);CHR$(129);"5...MPa"
540 PRINT TAB(0,15);CHR$(129);"6...in Hg"
550 PRINT TAB(0,16);CHR$(129);"7...mm Hg or torr"
560 PRINT TAB(0,17);CHR$(129);"8...m H2O"
570 PRINT TAB(0,18);CHR$(129);"9...dbar"
580 PRINT TAB(0,19);CHR$(129);"0...to exit to menu"
590 REM Disable cursor
```

```
600 VDU 23,1,0;0;0;0;
610 G$=GET$
620 PRINT TAB(15,22):CHR$(130)"WAIT"
630 REM Enable RS423 as output device
640 *FX3,7
650 IF G$="1"THEN PRINT"*0100EW*0100UN=1"
660 IF G$="2"THEN PRINT"*0100EW*0100UN=2"
670 IF G$="3"THEN PRINT"*0100EW*0100UN=3"
680 IF G$="4"THEN PRINT"*0100EW*0100UN=4"
690 IF G$="5"THEN PRINT"*0100EW*0100UN=5"
700 IF G$="6"THEN PRINT"*0100EW*0100UN=6"
710 IF G$="7"THEN PRINT"*0100EW*0100UN=7"
720 IF G$="8"THEN PRINT"*0100EW*0100UN=8"
730 IF G$="9"THEN PRINT"*0100EW*0100UN=0"
740 IF G$="0"THEN GOTO 810
750 FOR X=1TO 3000:NEXT
760 REM Enable screen as output
770 *FX3,4
780 GOTO 160
790 END
800 REM Enable keyboard as input
810 *FX2,2
820 REM Enable screen as output
830 *FX3,4
840 CHAIN"MENU"
>VDU3
```

>LIST

```
10 REM      ---PRESREM---
20 REM      17/1/91-S.KEENE
30 REM Pressure averaging program
40 REM for the Digiquartz standard
50 REM      via the RS423 port.
60 REM      USING BBC basic.
70 REM      Use "PRES" as the
80 REM      working program.
90 REM.
100 REM.
110 REM Exit to menu on escape
120 ON ERROR GOTO 910
130 REM Disable cursor
140 VDU 23,1,0;0;0;0;
150 D=0:N=0:S=0:E=0
160 DIM D$(11)
170 REM 1200 Baud receive
180 *FX7,4
190 REM 1200 Baud transmit
200 *FX8,4
210 REM Flush all buffers
220 *FX15,0
230 REM Enable RS423 as output device
240 *FX3,7
250 REM Clear Digiquartz command
260 PRINT"*0100VR"
270 REM Flush all buffers
280 *FX15,0
290 REM Enable RS423 as input device
300 *FX2,1
310 REM Printer sink to buffer input
320 *FX5,0
330 REM Show pressure units command
340 PRINT"*0100UN"
350 REM Enable printer as buffer
360 *FX3,2
370 INPUT A$
380 REM Enable screen as output
390 *FX3,4
400 CLS
410 PRINT TAB(11,4);CHR$(129);"PRESSURE IN:"
420 IF A$="*0001UN=1"THEN PRINT TAB(24,4);CHR$(134);"psi"
430 IF A$="*0001UN=2"THEN PRINT TAB(24,4);CHR$(134);"mbar or hPa"
440 IF A$="*0001UN=3"THEN PRINT TAB(24,4);CHR$(134);"bar"
450 IF A$="*0001UN=4"THEN PRINT TAB(24,4);CHR$(134);"kPa"
460 IF A$="*0001UN=5"THEN PRINT TAB(24,4);CHR$(134);"MPa"
470 IF A$="*0001UN=6"THEN PRINT TAB(24,4);CHR$(134);"in Hg"
480 IF A$="*0001UN=7"THEN PRINT TAB(24,4);CHR$(134);"mm Hg or torr"
490 IF A$="*0001UN=8"THEN PRINT TAB(24,4);CHR$(134);"m H2O"
500 IF A$="*0001UN=0"THEN PRINT TAB(24,4);CHR$(134);"dbar"
510 PRINT TAB(14,8);CHR$(129);"PRESSURE:"
520 PRINT TAB(10,12);CHR$(129);"THIS AVERAGE:"
530 PRINT TAB(10,14);CHR$(129);"LAST AVERAGE:"
540 PRINT TAB(0,18);CHR$(129);"DIFFERENCE IN AVERAGES:"
550 PRINT TAB(0,22);CHR$(129);"PRESS ESCAPE TO RETURN TO THE MENU"
560 REM Flush all buffers
570 *FX15,0
580 REM Enable RS423 as input device
590 *FX2,1
```

```
600 REM Enable screen as output
610 *FX3,7
620 REM Clear Digiquartz command
630 PRINT"*0100VR"
640 REM Flush all buffers
650 *FX15,0
660 REM Transmit pressure command
670 PRINT"*0100P4"
680 FOR W=0 TO 11
690 REM Enable printer as buffer
700 *FX3,2
710 REM Printer sink to buffer input
720 *FX5,0
730 INPUT D$(W):C$=MID$(D$(W),6)
740 REM Enable screen as output
750 *FX3,4
760 PRINT TAB(24,8) " "
770 PRINT TAB(24,8);CHR$(134);C$:NEXT
780 REM 790-890
790 S=0:F=0
800 FOR Q=0 TO 11
810 P$=RIGHT$(D$(Q),6):N=VAL(P$):S=S+N:NEXT:F$=STR$(S/12)
820 D=VAL(F$):PRINT TAB(24,12) " "
830 PRINT TAB(24,12);CHR$(134);D
840 PRINT TAB(24,14) " "
850 PRINT TAB(24,14);CHR$(134);E
860 F=D-E:E=D
870 PRINT TAB(24,18) " "
880 PRINT TAB(24,18);CHR$(134);F
890 GOTO 680
900 REM Flush all buffers
910 *FX15,0
920 REM Enable RS423 as output device
930 *FX3,7
940 REM Clear Digiquartz command
950 PRINT"*0100VR"
960 REM Enable screen as output
970 *FX3,4
980 PRINT TAB(13,23);CHR$(130);"WAIT"
990 REM Enable keyboard as input
1000 *FX2,2
1010 CHAIN"MENU"
>VDU3
```

> LIST

```
10 REM      ----TEMPREM----
20 REM      17/1/91-S.KEENE
30 REM.
40 REM Sampling temperature with
50 REM the Digiquartz standard
60 REM via the RS423 port.
70 REM Using BBC basic.
80 REM Use "TEMP" as the
90 REM working program
100 REM.
110 REM.
120 REM Exit to menu on escape
130 ON ERROR GOTO 470
140 REM Disable cursor
150 VDU 23,1,0;0;0;0;
160 CLS
170 PRINT TAB(5,4);(CHR$(129));"TEMPERATURE IN DEGREES C."
180 PRINT TAB(3,18);CHR$(129);"PRESS ESCAPE TO RETURN TO MENU"
190 REM Flush all buffers
200 *FX15,0
210 REM 1200 Baud receive
220 *FX7,4
230 REM 1200 Baud transmit
240 *FX8,4
250 REM Enable RS423 as output device
260 *FX3,7
270 REM Clear Digiquartz command
280 PRINT"*0100VR"
290 REM Flush all buffers
300 *FX15,0
310 REM Enable RS423 as input device
320 *FX2,1
330 REM Transmit temperature command
340 PRINT"*0100Q4"
350 REM Printer sink to buffer input
360 *FX5,0
370 REM Enable printer as buffer
380 *FX3,2
390 INPUT T$
400 C$=RIGHT$(T$,6)
410 REM Enable screen as output
420 *FX3,4
430 PRINT TAB(13,11);CHR$(134);C$
440 GOTO 380
450 END
460 REM Enable RS423 as output
470 *FX3,7
480 REM Flush all buffers
490 *FX15,0
500 REM Clear Digiquartz command
510 PRINT"*0100VR"
520 REM Enable screen as output
530 *FX3,4
540 PRINT TAB(13,20);CHR$(130);"WAIT"
550 REM Enable keyboard as input
560 *FX2,2
570 CHAIN"MENU"
```

>VDU3

